Code 83-24374-Opt3: 

“Apply a Route Layer's Definition Query to a Route Event layer”-ArcGIS 8.3 only
Public Sub CopyRouteLayerDefQuery(pRouteLayer As IFeatureLayer, pEventLayer As IFeatureLayer)

  On Error GoTo eh

  '+++ Sanity checks:

  '+++ (1) Make sure the route layer is truly a route layer.

  '+++ (2) Make sure the passed in event layer is truly an event layer

  If Not IsRouteLayer(pRouteLayer) Then

    MsgBox "Not a route layer", vbExclamation

    Exit Sub

  End If

  If Not IsRouteEventLayer(pEventLayer) Then

    MsgBox "Not an event layer", vbExclamation

    Exit Sub

  End If

  '+++ See if the route layer already has a definition query. Is it does not,

  '+++ then there is no point in continuing

  Dim pDef As IFeatureLayerDefinition

  Set pDef = pRouteLayer

  Dim sDefExp As String

  sDefExp = pDef.DefinitionExpression

  If Not Len(sDefExp) > 0 Then

    'MsgBox "Route layer has no definition query.", vbExclamation

    Exit Sub

  End If

  '+++ Get the event layer's route locator. If it already has a where clause specified

  '+++ ask the user if they want to overwrite it with the route layer's def query

  Dim pRES As IRouteEventSource

  Dim pRtLoc As IRouteLocator

  Dim Response

  Set pRES = pEventLayer.FeatureClass

  Set pRtLoc = pRES.RouteLocator

  If Len(pRtLoc.RouteWhereClause) > 0 Then

    Response = MsgBox("Event layer's route locator object already has a RouteWhereClause defined." & vbCrLf & _

      "Do you want to overwrite it?", vbYesNo + vbExclamation)

    If Response = vbNo Then Exit Sub

  End If

  '+++ We need to replace the event layer's route event source object with a new one. The new one

  '+++ will have a route locator that has an appropriate RouteWhereClause set.

  Dim pDS As IDataset

  Dim pRESName As IRouteEventSourceName

  Dim pName As IName

  Dim pRtLocName As IRouteLocatorName

  Set pDS = pRES

  Set pRESName = pDS.FullName

  Set pRtLocName = pRESName.RouteLocatorName

  pRtLocName.RouteWhereClause = sDefExp 'where we apply the route's def'n query

  Set pName = pRESName

  Set pEventLayer.FeatureClass = pName.Open 'this actually creates a new route event source object

  Exit Sub

eh:

  MsgBox Err.Number & ": " & Err.Description, vbCritical, "CopyRouteLayerDefQuery"

End Sub

Private Function IsRouteEventLayer(pFeatureLayer As IFeatureLayer) As Boolean

    IsRouteEventLayer = TypeOf pFeatureLayer.FeatureClass Is IRouteEventSource

End Function

Private Function IsRouteLayer(pFeatureLayer As IFeatureLayer) As Boolean

 '+++ Check if it is is polyline and it has measures

  Dim i As Long

  Dim pFC As IFeatureClass

  Dim pFlds As IFields

  Dim pGDef As IGeometryDef

  Set pFC = pFeatureLayer.FeatureClass

  '+++ We do not want to treat line event layers as route layers

  If TypeOf pFC Is IRouteEventSource Then

    IsRouteLayer = False

    Exit Function

  End If

  '+++ Make sure it is polyline

  If Not pFC.ShapeType = esriGeometryPolyline Then

    IsRouteLayer = True

    Exit Function

  End If

  '+++ Make sure it has measures

  Set pFlds = pFC.Fields

  i = pFlds.FindField(pFC.ShapeFieldName)

  Set pGDef = pFlds.Field(i).GeometryDef

  IsRouteLayer = pGDef.HasM

End Function

